Sage Days 37

Korean Mathematics Society
Spring Meeting

Prof. Robert Beezer
University of Puget Sound
Tacoma, Washington USA

April 28, 2012

An Example from Discrete Mathematics

Sage has many graphs built-in.

```
graphs.
```

We will experiment with the Heawood graph.

```
G = graphs.HeawoodGraph()
```

```
G.plot()
```

Smallest degree 3 graph with no circuit of length 5 or less.

```
G.is_regular()
```

```
G.degree()
```

```
G.girth()
```

A bipartite graph.

```
G.chromatic_number()
```

The color classes of a 2-coloring, and an improved plot.

```
classes = G.coloring()
classes
```

```
G.plot(partition=classes, vertex_size=500, thickness=4)
```

Group Theory

Edge-preserving permutations of the vertices.

```
A = G.automorphism_group()
A
```

```
A.order()
```

Graph is "vertex-transitive" since the permutation group is transitive.

```
A.orbits()
```

The automorphism group of this graph is the projective general linear group $PGL(2,7)$.

```
PGL27 = PGL(2,7)
PGL27
```

```
PGL27.is_isomorphic(A)
```

Properties of this group.

```
A.is_simple()
```

```
[c.order() for c in A.composition_series()]
```

The group of order 168 is the second smallest nonabelian simple group, $PSL(2,7)$.

Designs

The Fano plane, a combinatorial structure with a simple automorphism group. Also known as a 2–$(7,3,1)$ design.



```
fano = BlockDesign(7,[[0,1,2],[0,3,4],[0,5,6],[1,3,5],[1,4,6],[2,3,6],[2,4,5]]
fano
```

```
F = fano.automorphism_group()
F.order()
```

```
F.is_simple()
```

```
PSL27 = PSL(2,7)
PSL27.is_isomorphic(F)
```

Linear Algebra

```
M = G.adjacency_matrix()
M
```

"fcp()" is the factored characteristic polynomial.

```
M.fcp()
```

```
ev = M.eigenvalues()
ev
```

Field Extensions, Interval Arithmetic

```
rho = ev[3]
rho, rho^2

rho^2 == 2

rho.minpoly()

info = rho.as_number_field_element()
info

N = info[0]; N

N.degree()

N.base_field()

info[1], N.gens()
```

We can get greater precision in an "interval field"

```
RIF128 = RealIntervalField(128)
rho.interval(RIF128)
```

This worksheet available at:
`http://buzzard.ups.edu/talks.html`